
2-17

Code Accessing Pre- and Post-Simulation Values of Coefficients

DECOMP.TAB also needs to access pre-simulation values of Coefficients (for example BVOA and
BVOM below) and post-simulation values of the same Coefficients (for example, UVOA and UVOM
below). These are calculated using values of things like BVFM (base values) and UVFM (updated
values) which are read from files GTAPBASE and GTAPUPDATE respectively.

Code in DECOMP.TAB

File  GTAPBASE # file containing all base data #;
File  GTAPUPDATE # file containing all updated base data #;
Coefficient (all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG) BVFM(i,j,r)
   # Base producer expenditure on i by j in r at market prices #;
Read      BVFM from file GTAPBASE header "VFM";
Coefficient (all,i,NSAV_COMM)(all,r,REG)  BVOM(i,r)
   # Base value of commodity i output in region r at mkt prices #;
Formula (all,i,ENDW_COMM)(all,r,REG)
    BVOM(i,r) = sum(j,PROD_COMM, BVFM(i,j,r));
! Then formulas for BVOM(i,r) for i in other sets (MARG_COMM etc) !
! Similarly for BVOA(i,r) – Base values of VOA !
Coefficient (all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG) UVFM(i,j,r)
  # Updated producer expenditure on i by j in r at market prices #;
Read     UVFM from file GTAPUPDATE header "VFM";
Coefficient (all,i,NSAV_COMM)(all,r,REG)  UVOM(i,r)
   # Updated value of commodity i output in region r at mkt prices #;
Formula (all,i,ENDW_COMM)(all,r,REG)
    UVOM(i,r) = sum(j,PROD_COMM, UVFM(i,j,r));
! Then formulas for UVOM(i,r) for i in other sets (MARG_COMM etc) !
! Similarly for UVOA(i,r) – Updated values of VOA !
! Then are formulas below calculating pre-sim tax rates BOUTAX
   and post-sim tax rates UOUTAX using BVOA,BVOM and UVOA,UVOM !
Zerodivide (nonzero_by_zero) default 0;
Coefficient (all,i,NSAV_COMM)(all,r,REG)  BOUTAX(i,r);
Formula (Initial) (all,i,NSAV_COMM)(all,r,REG)
    BOUTAX(i,r) = [BVOM(i,r) - BVOA(i,r)] / BVOM(i,r) * 100;
Coefficient (all,i,NSAV_COMM)(all,r,REG)  UOUTAX(i,r);
Formula (all,i,NSAV_COMM)(all,r,REG)
    UOUTAX(i,r) = [UVOM(i,r) - UVOA(i,r)] / UVOM(i,r) * 100;
Zerodivide (nonzero_by_zero) off;
Coefficient (all,i,NSAV_COMM)(all,r,REG)(all,c,COL) OUTPUT(i,r,c) ;
Formula (all,i,NSAV_COMM)(all,r,REG)  OUTPUT(i,r,"taxrateb") = BOUTAX(i,r);
Formula (all,i,NSAV_COMM)(all,r,REG)  OUTPUT(i,r,"taxrateu") = UOUTAX(i,r);
Write OUTPUT to file WELVIEW header "A211"
 longname "combination of output welcnt,vol change and tax rate";

Using Release 9 software, this could all be done in GTAP.TAB. The Formula for BOUTAX must be
done as a Formula(Initial) in a non-postsim part. The right-hand side of this Formula can refer to VOM
and VOA (since these are equal to the Base values in a Formula(Initial) in a non-postsim part). There is
no need for Coefficients BVOA and BVOM to be declared explicitly. Of course the Formula(Initial)
for BOUTAX must be placed after the values of Coefficients VOA and VOM have been calculated via
the usual Formulas in GTAP.TAB. Nor would there be any need for the Coefficients UVOA and
UVOM to be declared explicitly (and hence no need of File GTAPUPDATE). If VOA and VOM are
accessed in a post-simulation part of the code (see the Formula for UOUTAX below), their values are
automatically the post-simulation (or, updated) values. That is, the following code could be added to
GTAP.TAB to achieve what is done in the section of DECOMP.TAB code shown above. Note that the
first part (calculating BOUTAX) must not be in a post-sim part while the second part (calculating
UOUTAX) must be in a post-sim part.

Code which could be added to GTAP.TAB

! These formulas setting base values do not go in a post-sim part !
 Zerodivide (nonzero_by_zero) default 0;
Coefficient (Parameter) (all,i,NSAV_COMM)(all,r,REG) BOUTAX(i,r)
  # Base value of output tax rates # ;
Formula (Initial) (all,i,NSAV_COMM)(all,r,REG)
! VOM,VOA here are base values since is Formula(Initial) !
    BOUTAX(i,r) = [VOM(i,r) - VOA(i,r)] / VOM(i,r) * 100;
Zerodivide (nonzero_by_zero) off;

! Next must be in a post-sim part !
PostSim (Begin) ;
Zerodivide (nonzero_by_zero) default 0;
Coefficient (all,i,NSAV_COMM)(all,r,REG) UOUTAX(i,r)
  # Updated value of output tax rates # ;
Formula (all,i,NSAV_COMM)(all,r,REG)
! VOM,VOA here are the updated values since in post-sim part !
    UOUTAX(i,r) = [VOM(i,r) - VOA(i,r)] / VOM(i,r) * 100;
Zerodivide (nonzero_by_zero) off;
Coefficient (all,i,NSAV_COMM)(all,r,reg)(all,c,col) OUTPUT(i,r,c) ;
Formula (all,i,NSAV_COMM)(all,r,reg)
   OUTPUT(i,r,"taxrateb") = BOUTAX(i,r);
Formula (all,i,NSAV_COMM)(all,r,reg)
   OUTPUT(i,r,"taxrateu") = UOUTAX(i,r);
Write OUTPUT to file WELVIEW header "A211"
 longname "combination of output welcnt,vol change and tax rate";
PostSim (End) ;

GTAPVOL.TAB

Another part of the post-simulation processing supplied with the GTAP model is GTAPVOL.TAB.
This has code similar to the above examples from DECOMP.TAB. All of the processing done via
GTAPVOL.TAB  could be done more simply in post-simulation parts of GTAP.TAB.

2.3.2 Other Models

Sophisticated post-simulation processing could be set up easily for other models in the main TAB file
for those models. Then you won’t need a fancy Windows interface such as RunGTAP for your model.
Every time you carry out a simulation, the post-solution processing will be done at the same time. We
say a little more about this in section 2.4 below.




